

10

Stock Take

Here is a program with thousands of applications. In the home, the workshop or the office, use it for keeping an inventory of ingredients, equipment, materials or parts. Not only does it display a complete list, it will prepare a list of all items for which present stocks are below the minimum level. Deciding what you need to buy to maintain your stock is just a matter of noting down the details from the screen as the Electron presents them to you. If your stock is of a perishable kind, such as the smoked salmon in the deep-freeze, or the packets of photographic paper on your dark-room shelf, the program can identify old stock by displaying a list of all items entered before a selected date.

STOCK TAKE is a tape-based program. This means that several people can use the program, each having their own tape or tapes on which their stock lists are held. There can be a tape for the cook, listing the contents of the food cupboard and the deep-freeze, and a components tape for the electronic wizard of the household. The gardener of the family can have a tape for stocks of fertilizer, seeds, flowerpots and other consumable items. The home brewer can list stocks of sugar, malts, hops, dried fruits and flowers, as well as of the finished products. The DIY expert can list stocks of timber, nails, screws and all the countless other items needed in the efficient home workshop. It is very useful to keep an inventory of the furniture and furnishings for insurance purposes. The program can hold details of dates of purchase and the cost of all such items, making it very easy to be certain that your insurance cover is adequate year by year.

Although the program has many features built in, and is ready to use, it is designed so that it can be tailored to your requirements, as explained in the *Variations* section of this chapter.

There is room for up to 200 records on any one stock list.

STOCK TAKE records

Each record on the stock list consists of eight items of information:

(1) *Code*: a word up to four letters long, or a group of up to four letters and numerals. The purpose of this is to identify certain categories of stock quickly. For example, a stock list of electronic components could have the code RES for all items that referred to stocks of resistors. In a kitchen stock list, you might use the code DF for all deep-freeze stock, to distinguish it from stocks of other kinds. Keying in DF would then make the computer display a list of all deep-frozen stocks, as explained later.

(2) *Description*: words or codes to identify each item exactly. Up to ten characters are allowed here. This number has been restricted in order to fit a complete record into one line of a 40-column screen. However, it is usually easy to shorten words or use spare parts numbers so as to specify the nature of the item exactly within the space allowed.

(3) *Date*: this can be the date of purchase, the date on which the stock quantity was last amended, or any other date which suits your system of stock-taking. You can enter the date in any format you choose, provided that it has no more than six characters. Thus the 3rd September 1984 could be keyed in as '030984', '30984', '090384', '03SE84', '3 SEPT', or '3/9'. The only point to remember is that the 'Old List' routine works only with dates entered according to the first format above (DDMMYY).

(4) *Quantity*: this is the number of items currently held in stock. The program accepts any number (including decimal fractions) up to 9999. It is best not to use numbers with more than two places of decimals, for this upsets the arrangement of the display. The same points apply to numbers entered in items (5) and (6) below.

(5) *Minimum Quantity*: this is the minimum quantity that you wish to keep in stock. When the stock quantity has fallen to this minimum or below it, you need to buy more of the item. The 'Order List' option (see later) tells you what items have to be replenished.

(6) *Re-order quantity*: this is the quantity of each item which should be bought when the stock quantity falls below the minimum level.

(7) and (8) *Fields 1 and 2*: these give you two spare columns in the stock list in which you can record any information you wish. One of the fields might give the latest price of the item; another might say from whom the item was purchased.

Using the program

Before you use the program for the first time, or when you want to begin a new stock list, you need to prepare a tape on which the stock list is to be recorded. Use the START LIST program for this. All this does is to put an 'emptylist on to tape, ready for use by the STOCK TAKE program.

Run the START LIST program. When it asks you 'NAME OF LIST,?' type in the name (up to 7 characters) you have decided to give to your stock list. It might be 'CANFOOD' 'STATNRY' 'GARDEN' or 'SPARES'. Write the chosen name on the cassette label, then put the cassette into the tape recorder. The screen will show the message 'RECORD then RETURN'. Wind the tape to the beginning, if necessary, so that it is ready for making a recording. Press the RECORD and PLAY buttons on the recorder, then press RETURN on the Electron. Creating the 'emptylist will take only a few seconds. Then the message 'Emptylist ready' appears on the screen, and the program has finished.

Now load the STOCK TAKE program. When you RUN this program, you are first asked to key in the name of the list. This is the name you gave it when you used the START LIST program. Key in its name, and press RETURN. Remove the STOCK TAKE tape and replace it with the 'listtape. Rewind the tape to the beginning, if necessary, then press the PLAY button on the recorder. If you are starting a new list, loading it takes only a few seconds. If, on the other hand, you are loading a list which already holds many items, it will take several minutes.

When loading is complete, the screen clears and the menu is displayed. This allows you a choice of any of six options:

(1) *New Entry*: Use this to take a new item on to the list. You are asked to key in the eight pieces of information required. Refer to the headings (under STOCK TAKE records) to remind yourself of these, and what restrictions (e.g. number of characters allowed) each has. Press RETURN after entering each piece of information. You do not have to make an entry for each heading. If there is nothing to enter, just press RETURN. When entry is complete, the screen clears and displays everything you have just typed, in the form in which it will be taken on to the list. This gives you a chance to look for typing errors.

Look to see if you have typed in more than the allowed number of characters, and so have ' lost the end of an entry. If all is correct, key ' Y' and you will be returned to the menu. Otherwise, type ' N' to cancel the entry and re-enter the details correctly.

There is no indication in the program of the number of records currently in the list. Up to 200 records are allowed. When the number held exceeds 195 the Electron ' beeps' at you each time you choose this option. If there are more than 200 records, it returns you to the menu.

(2) *Display stock list*: This shows you the whole list from beginning to end, in the order in which the records were entered. Figure 10. 1 is an example of a list compiled by a DIY enthusiast. The items are coded, T standing for timber, N for nails, S for screws, PE for emulsion paint and so on. Alternate columns of information are displayed on the screen in red-on-yellow, or yellow-on-red, to make them easy to distinguish when packed with information. If you have more than a screenful of records, the Electron waits for you to read them. Press SHIFT to view the next screenful. This is a general rule for this program; if the display halts, press SHIFT. Occasionally it may stop with the bottom line of the screen showing ' Spacebar = continue' . Pressing the space-bar will have no effect until after you have pressed SHIFT.

When the whole list has been displayed, the message at the bottom of the screen is ' Spacebar = Continue' Pressing the space-bar returns you to the menu.

(3) *Display/Amend entry*: This option finds any single entry or group of similar entries without the need to view the whole stock list. An entry may be defined by specifying its code or its description. You are asked to type in both of these, but need type in only one. Press RETURN. If you specify the code, you will be shown only the records which have this code. For example, with the list shown in Fig. 10.1 , the code ' T' would cause all entries relating to timber to be displayed. If you type a description only, items having that description will be displayed. Since each description should be unique, you will probably be shown only one item.

The records are displayed one at a time to give you the opportunity of deciding whether to amend them. If you do not want to amend a record, press the space-bar. If there are further records conforming to the code or description, the next one will then be displayed. If there are no more, the message ' ENDOF SEARCH' is displayed. Press the

CODE	DESCRIPTION	DATE	QUANTITY	MIN.QTY	RE-ORDER	FIELD 1	FIELD 2
T	PINE 3X3	030184	6	2	6		
T	PINE 8X2	060184	1	2	6		
T	CONTI 9	250184	3	1	2	WH	
N	WIRE 2.5	010284	100	20	50		
N	WIRE 5	010284	100	20	50		
N	WIRE 15	010284	20	10	10		
S	ST CS 4/5	010284	25	10	25		
AD	ARALDITE R	010284	1	1	1	*	
V	RONSEAL OD	050284	500	500	500		ml
S	BR RH 4/5	120284	20	10	10		
F	CUPHK B S	130384	6	2	3	RND	
F	CUPHK BR L	050384	3	2	3	SQ	
TOOL	HKSAW BL	260384	1	1	2	SD	
N	PP 1 B	100484	2	1	1		PKT
F	HNGE B 5	170484	1	4	2		
PE	WHITE	210485	5	1	5	*	1
PE	LILAC	210484	3	1	2	*	1
SO	WHT SPRT	010584	200	100	500	*	
TOOL	HKSAW BL	100584	2	4	10		
S	ST HCS 2/2	090684	4	10	10		
TOOL	SURFORM S	090784	1	2	1	WW	
M	STAPLE 4	190784	50	10	25	GALV	
T	OAK 4X4	260884	3	1	1		SA
PG	SCARLET	010984	100	250	250		ml
PG	GOLD	030984	1	1	1	CAN	
M	POLYSTRIPP	140984	1	1	1	BOT	
T	7-PLY 10	110984	0	1	1	SHT	
S	ST CS 3/4	101184	12	20	20	SD	WW
SO	METHS	121184	500	100	500	*	ml
AD	BOSTIK 1	111284	1	1	1	TUBE	

Fig. 10.1. Complete stock list

space-bar to return to the menu.

If you decide to amend a record you can either amend the figure in the 'Quantity' column only (key 'Q' or one or more of the other details (key 'A' Typing 'Q' provides a quick way of altering the quantities in records as the stock decreases, or when new purchases are made. When you amend only the stock quantity, the screen clears, and you are reminded what the present quantity is. You are then asked to key in the new quantity. After you have pressed RETURN, you will be shown the next record conforming to your specification, or 'END OF SEARCH'.

If you decide to amend other parts of the record, you are asked to follow the same procedure as under option 1. All details have to be typed in again, with corrections and amendments. The original version of the record remains on the screen while you type, to allow you to copy in details that are not to be changed. This option can be used to replace an unwanted record with a completely new one.

(4) *Order list*: Selecting this option makes the computer scan the stock list, displaying all records for which the quantity of stock is equal to or less than the minimum stock quantity. Figure 10.2 shows the Order List produced from the stock list of Fig. 10.1. If there are many such records, they are displayed by the screenful. Press SHIFT to view the next screenful.

(5) *Old list*: Before the computer searches the list, you are asked to key in the date to be used in the search. This could be the date of purchase of stock, or the 'best before' date of stored foodstuffs. The date is specified by typing in the month (1—12) and the year (0—99). The computer then looks through the list and displays all records for the months up to and including the month specified. Figure 10.3 shows the Old List obtained by selecting February 1984 as the date. The Variations section explains how to modify the program to work with day-and-month dates instead of month-and-year dates. This would be more suitable for stores such as foodstuffs, which have a relatively short shelf-life.

As with item 4, the Old list is displayed by screenfuls. Press SHIFT to view the next screenful.

(6) *Finish*: This option must be used every time you have finished a session with the program, in which you have used options 1 or 2 to start new records, or to alter existing ones. As soon as you have

CODE	DESCRIPTION	DATE	QUANTITY	MIN.QTY	RE-ORDER	FIELD 1	FIELD 2
T	PINE 8X2	060184	1	2	6		
AD	ARALDITE R	010284	1	1	1	*	
V	RONSEAL OD	050284	500	500	500		ml
TOOL	HKSAW BL	260384	1	1	2	SD	
F	HNGE B 5	170484	1	4	2		
TOOL	HKSAW BL	100584	2	4	10		
S	ST HCS 2/2	090684	4	10	10		
TOOL	SURFORM S	090784	1	2	1	WW	
PG	SCARLET	010984	100	250	250		ml
PG	GOLD	030984	1	1	1	CAN	
M	POLYSTRIPP	140984	1	1	1	BOT	
T	7-PLY 10	110984	0	1	1	SHT	
S	ST CS 3/4	101184	12	20	20	SD	WW
AD	BOSTIK 1	111284	1	1	1	TUBE	

Fig. 10.2. Order list.

CODE	DESCRIPTION	DATE	QUANTITY	MIN.QTY	RE-ORDER	FIELD 1	FIELD 2
T	PINE 3X3	030184	6	2	6		
T	PINE 8X2	060184	1	2	6		
T	CONTI 9	250184	3	1	2	WH	
N	WIRE 2.5	010284	100	20	50		
N	WIRE 5	010284	100	20	50		
N	WIRE 15	010284	20	10	10		
S	ST CS 4/5	010284	25	10	25		
AD	ARALDITE R	010284	1	1	1	*	
V	RONSEAL OD	050284	500	500	500		ml
S	BR RH 4/5	120284	20	10	10		

Fig. 10.3. Old list.

selected option 6, the screen clears. The message ' RECORD then RETURN' appears. Rewind the data tape to the place at which the recorded list begins. Press the ' RECORD and ' PLAY buttons on the tape recorder. Then press RETURN on the Electron. Alonglist may take several minutes to save. When this is complete, the message ' BACKUP appears. To take a backup copy, key ' Y and repeat the procedure above. If you press ' N the message ' STOCKTAKE FINISHED' is displayed.

Keying in

The string which is listed by our printer as a row of left-arrows (line 170) is a row of ' underline characters. Press SHIFT and the ' _ down-arrow' key to obtain this. Note the spaces on either side of each column heading in line 1220, except for CODE. The sets of double-quotes in lines 1310 and 1320 all have four spaces between them, except for the second set on line 1310, which has ten spaces.

Program design

20-40 Initialising Mode, arrays and colours.
 50-150 Loading data from tape.
 160-270 Menu and directing to options.
 280-300 New Entry routine.
 310-380 Display stock list.
 390-600 Display/Amend entry.
 610-720 Order list.
 730-770 Old list.
 780-920 Finish.
 930--950 PROCyn to accept ' Y' or ' N' .
 960-1200 PROCentry for inputting data and confirmiig it.
 1210-1230 PROCheadings td display column headings.
 1240-1260 PROCry to display red text on yellow background.
 1270-1290 PROCyr to display yellow text on red background.
 1300-1340 PROClist to display details of one record in the stocklist.
 1350-1380 PROCspace to wait for space-bar.

The program (START LIST)

```

10 REM ** START LIST **
20 CLS:INPUT'"NAME OF LIST "T$
30 X=OPENOUT T$
40 PRINT #X,0:CLOSE #X
50 PRINT'"Empty list ready"
60 END

```

The program (STOCK TAKE)

```

10 REM ** STOCK TAKE **
20 MODE4
30 DIM D$(200,4),D%(200,2):VDU14
40 VDU 19,1,1,0,0,0:VDU 19,0,3,0,0,0
50 CLS:INPUT'"NAME OF LIST "T$
60 T$=LEFT$(T$,7)
70 X=OPENIN T$
80 INPUT#X,N
90 IF N=0 THEN 150
100 FOR J=1 TO N
110 FOR K=0 TO 4
120 INPUT#X,D$(J,K):NEXT
130 FOR K=0 TO 2
140 INPUT#X,D%(J,K):NEXT:NEXT
150 CLOSE#X
160 CLS:PRINT'TAB(11)"**STOCK TAKE**"
170 PRINTTAB(13)"[[[[[[[[[[["
180 PRINT'"TAB(7)"1) NEW ENTRY"
190 PRINT'TAB(7)"2) DISPLAY STOCK LI
ST"
200 PRINT'TAB(7)"3) DISPLAY/AMEND EN
TRY"
210 PRINT'TAB(7)"4) ORDER LIST"
220 PRINT'TAB(7)"5) OLD STOCK LIST"
230 PRINT'TAB(7)"6) FINISH"
240 PRINT'"'"<SELECT BY PRESSING NU
MBERED KEY>"
250 REPEAT:KEY$=GET$
260 A=VAL(KEY$):UNTIL A>=1 AND A<=6
270 CLS:ON A GOTO 280,320,390,610,730,
780
280 N=N+1:J=N:IF N>195 THEN PRINT CHR$

```

132 Practical Programs for the Electron

```
(7):IF N>200 THEN N=200:GOTO 160
  290 PROCentry
  300 GOTO 160
  310 CLS:PRINT';T$;" STOCK LIST"
  320 PROCheadings
  330 FOR J=1 TO N
  340 PROClist
  350 NEXT:PROCry
  360 PRINT'''' < SPACE BAR = CONTIN
UE >"
  370 REPEAT:KEY$=GET$: UNTIL KEY$=" "
  380 GOTO 160
  390 CLS:PRINT'"DISPLAY ENTRY"
  400 INPUT'"CODE "C$
  410 INPUT'"DESCRIPTION "D$
  420 CLS:PROCheadings
  430 FOR J=1 TO N
  440 IF C$<>D$(J,0) AND D$<>D$(J,1) THE
N 530
  450 PROClist
  460 PRINT'''' < SPACE=CONTINUE : Q=QT
Y : A=AMEND >":PRINT
  470 REPEAT:KEY$=GET$:UNTIL KEY$=" " OR
KEY$="Q" OR KEY$="A"
  480 IF KEY$=" "THEN 530
  490 IF KEY$="Q"THEN 570
  500 PRINT':PROCentry
  510 CLS:PROCheadings
  520 PROClist
  530 NEXT:PROCry
  540 PRINT'''TAB(10)"**END OF SEARCH**"
  550 PROCspace
  560 GOTO 160
  570 CLS:PRINT'"QUANTITY IS: ";D$(J,0)
  580 INPUT'"ALTER TO "D:IF D>9999 THEN
580
  590 D$(J,0)=D
  600 GOTO 530
  610 CLS:PROCheadings
  620 FOR J=1 TO N
  630 IF A=4 AND D$(J,0)>D$(J,1) THEN 69
0
  640 IF A<>5 THEN 660
  650 YR=VAL(RIGHT$(D$(J,2),2)):IF YR>Y
```

```

OR Y=YR AND M<VAL(MID$(D$(J,2),3,2)) THEN
690
    660 REM** OTHER SEARCHES **
    670 REM** OTHER SEARCHES **
    680 PROClist
    690 NEXT:PROCry
    700 PRINT' ' ' ' < SPACE BAR = CONTIN
UE > "
    710 REPEAT: KEY$=GET$: UNTIL KEY$=" "
    720 GOTO 160
    730 CLS:REPEAT:INPUT' ' "WHICH MONTH (1
- 12) "M$
    740 M=VAL(M$):UNTIL M>=1 AND M<=12
    750 REPEAT:INPUT' "WHICH YEAR (0 - 99)
"Y$
    760 Y=VAL(Y$):UNTIL Y>=0 AND Y<=99
    770 GOTO 610
    780 PRINT' ':X=OPENOUT T$
    790 PRINT#X,N
    800 FOR J=1 TO N
    810 FOR K=0 TO 4
    820 PRINT#X,D$(J,K)
    830 NEXT
    840 FOR K=0 TO 2
    850 PRINT#X,D%(J,K)
    860 NEXT:NEXT
    870 CLOSE#X
    880 PRINT' "BACK-UP? (Y/N)"
    890 PROCyn
    900 IF KEY$="Y" THEN 780
    910 PRINT' "STOCK TAKE FINISHED"
    920 END
    930 DEF PROCyn
    940 REPEAT:KEY$=GET$:UNTIL KEY$="Y" OR
KEY$="N"
    950 ENDPROC
    960 DEF PROCentry
    970 INPUT' ' "CODE "D$:D$(J,0)=LEFT$(D$,
4)
    980 INPUT"DESCRIPTION "D$:D$(J,1)=LEFT
$(D$,10)
    990 INPUT"DATE (DDMMYY) "D$:D$(J,2)=LE
FT$(D$,6)
    1000 INPUT"QUANTITY "D:IF D>9999 THEN 1

```

134 Practical Programs for the Electron

```
000
  1010 D%(J,0)=D
  1020 INPUT"MINIMUM STOCK "D:IF D>9999 T
HEN 1010
  1030 D%(J,1)=D
  1040 INPUT"RE-ORDER QUANTITY "D:IF D>99
99THEN 1020
  1050 D%(J,2)=D
  1060 INPUT"FIELD 1 "D$:D$(J,3)=LEFT$(D$
,4)
  1070 INPUT"FIELD 2 "D$:D$(J,4)=LEFT$(D$
,4)
  1080 CLS:PRINT'"DATA TO BE ENTERED IS:
"
  1090 PRINT'"CODE:";TAB(19)D$(J,0)
  1100 PRINT"DESCRIPTION";TAB(19)D$(J,1)
  1110 PRINT"DATE";TAB(19)D$(J,2)
  1120 PRINT"QUANTITY";TAB(19)D%(J,0)
  1130 PRINT"MINIMUM STOCK";TAB(19)D%(J,1
)
  1140 PRINT"RE-ORDER QUANTITY";TAB(19)D%
(J,2)
  1150 PRINT"FIELD 1";TAB(19)D$(J,3)
  1160 PRINT"FIELD 2";TAB(19)D$(J,4)
  1170 PRINT'"'TAB(22)"ALL OK? Y/N"
  1180 PROCyn
  1190 IF KEY$="N" THEN CLS:GOTO 960
  1200 ENDPROC
  1210 DEF PROCheadings
  1220 PROCry:PRINT'"CODE";:PROCyr:PRINT
" DESCRIPT ";:PROCry:PRINT" DATE ";:PROC
yr:PRINT" QY ";:PROCry:PRINT" MN ";:PROC
yr:PRINT" RE ";:PROCry:PRINT" F1 ";:PROC
yr:PRINT" F2 "
  1230 ENDPROC
  1240 DEF PROCry
  1250 COLOUR 1:COLOUR 130
  1260 ENDPROC
  1270 DEF PROCyr
  1280 COLOUR 2:COLOUR 129
  1290 ENDPROC
  1300 DEF PROClist
  1310 PROCry:PRINT;LEFT$(D$(J,0)+"      ",
4);:PROCyr:PRINT;LEFT$(D$(J,1)+"
```

```

    ,10)::PROCry:PRINT;D$(J,2)::PROCyr:PR
INT;RIGHT$( "      "+STR$(D%(J,0)),4)::PROC
ry:PRINT;RIGHT$( "      "+STR$(D%(J,1)),4);
1320 PROCyr:PRINT;RIGHT$( "      "+STR$(D%
(J,2)),4)::PROCry:PRINT;LEFT$(D$(J,3)+ "
    ",4)::PROCyr:PRINT;LEFT$(D$(J,4)+ "
    ",4);
1330 PROCry
1340 ENDPROC
1350 DEF PROCspace
1360 PRINTTAB(5)" < SPACE BAR TO CONTIN
UE > "
1370 REPEAT:KEY$=GET$:UNTIL KEY$=" "
1380 ENDPROC

```

Variations

One of the features of this program is that it is adaptable. The two spare items, Field I and Field 2 can be used to hold any kind of data. If you have decided on a definite use for these columns, change their headings by amending lines 1060, 1070, and 1220.

The Old List normally requires you to enter a month-and-year date. To change to a day-and-month date, make these changes in the program:

Alter the message in line 730 to "WHICH DAY (1-31)"

Alter line 740 to:

```
740 M=VAL(M$):UNTIL M>=1 AND M<=31
```

Alter the message in line 750 to "WHICH MONTH (1-12)"

Alter line 760 to:

```
760 Y=VAL(Y$):UNTIL Y>=1 AND Y<=12
```

Change line 650 to:

```
650 YR=VAL(MID$(D$(J,2),3,2)):IF YR>Y OR Y=YR AND
M<VAL(LEFT$(J,2),2) THEN 690
```

In Options 3 to 5, the program searches the stock list looking for

records having a certain feature. Examples are the date (Old List) and the relation between stock quantity and minimum re-order quantity (Order List). Lines 630 and 650 identify such records, and make the computer skip over any which do not conform to the specification. Space has been left in the program at lines 660 and 670 for you to write your own search routines. These could be used in conjunction with information in any column, including fields 1 and 2. You would need to add an extra option to the Menu, to allow the requirements to be keyed in. Then the program would go to line 610 for the search to be made and the list displayed. In this way you could make the computer display all records which refer to the same supplier, for example, or prepare a list of all items falling within a given price range. With suitable coding in fields 1 and 2, the program could display a list of foodstuffs needing special storage requirements, or those which are to be purchased in large quantities in certain seasons, for deep-freezing. If you do not wish to go to the trouble of writing extra routines, it may be possible to make use of the code in the first column of the list. If the code is used to indicate certain categories of item, Option 3 will allow you to find all of these, displaying them one at a time.