

1

Space Plan

One of the easiest ways of planning the layout of furniture in a room is to draw on paper an outline of the room and cut out paper shapes to represent the furniture. Then you can slide the shapes around on the plan until you find the most suitable arrangement. This is certainly less tiring than heaving the furniture itself around in the real room! This is also a good way to plan the layout of your garden, or of a workshop or office.

This program makes everything even easier. It works on a similar principle, except that it uses multicoloured displays on the computer screen. You draw an outline of the room or garden on the screen, using the computer's editing keys. Then you can plot circles, rectangles and any other shapes on the screen to represent the furniture or flowerbeds, paths and trees. They can be moved to different parts of the screen. They can be turned around to fit them into place, as required.

The program provides for plans to be saved on tape and loaded back into the computer for immediate display on a later occasion. This could be particularly helpful if you want to prepare several versions of a plan. You can draw a plan and save it. Then you can modify the plan, perhaps moving some of the shapes to different positions, or adding new shapes, and then save the modified version under its own file name.

As well as the applications mentioned above, SPACE PLAN can be used for planning layouts of anything from a stage set to the front page of a magazine. Dressmakers will find that it is easier to move the shapes around the screen than it is to arrange Himsy pattern-pieces on a length of fabric. Rearranging the pieces or adapting the designer's layout to fabric of different widths can often save money.

Finally, if you are keen on abstract art, why not use the colourful shapes to create your next masterpiece? Record on tape your own portfolio of computer art.

Using the program

The first message to appear on the screen asks you ' PLANSON TAPE? (Y/N)' If you already have a plan saved on tape and wish to load it and perhaps do some more work on it, then press ' Y' Otherwise press ' N' If you press ' N' the program continues as described under the heading Commands, below.

If you press ' Y' you will then be asked ' NAME OF FILE?' Key in the file name under which the plan has been saved and press RETURN. The recorder motor starts, allowing you to wind the tape (if necessary) to position it correctly. Then press PLAY on the recorder. In a few moments the loading is complete.

Commands

The screen clears to black, with a narrow light blue strip across the bottom. This strip is where messages appear during the running of the program. The black area of the screen is available for your space plan.

The first message to appear is ' BCDMPRSTZ?' This curious combination of letters informs you which keys will produce an effect at this stage. The effect of each key is as follows:

- B Bring back a shape that you have previously deleted.
- C Draw a circle.
- D Delete a shape from the screen.
- M Move a shape.
- P Draw a polygon.
- R Draw a rectangle.
- S Alter the scale of the plan.
- T Tidy the drawings.
- Z Save a plan on tape.

The meanings of these commands will become apparent from the descriptions below.

Drawing shapes

The three drawing commands are ' C' ,P and ' R' If you key ' C' (no need to press RETURN) a new message appears, asking ' Which one? (0-9)' You can draw up to 10 circles on the screen. Each circle has its

own number from 0 to 9. It is as well to note on paper, for future reference, what each circle is intended to represent. In a room plan, a circle might represent a circular table or rug, or perhaps a lamp. In a garden plan, a circle could represent a circular garden bed, a pool, or the crown of a tree. Decide what number the circle is to have and press one of the keys, 0 to 9.

The next message to appear asks ' Colour{1—7}' In response to this, press one of the keys 1 to 7, depending on what colour you want to use for the circle. The colours are numbered as for Mode 2:

- 1 = Red
- 2 = Green
- 3 = Yellow
- 4 = Blue
- 5 = Magenta
- 6 = Cyan
- 7 = White

Select the colour you require and press the key. Then the final message appears. This asks ' Whatradius?' In response to this you have to key in a number. Only positive integers (numbers without decimal places) are allowed, but there is no restriction on the size of the number. Normally it should be in the range 5 to 250, but larger and smaller values can be used. The number determines how big the circle is to be. When the program is first run, the width of the black drawing area from left to right is 640 screen units. If you are drawing a plan to scale, you can think of this width as equivalent to 640 centimetres (6.4 metres, the length of a large living-room), or 640 inches (the length of an average garden) or perhaps 64 metres (a large garden).

The height of the drawing area is 480 units, though you may not have the whole of this available if your TV does not show the top line of the Electron's display.

As soon as you have keyed in the radius, press RETURN. The circle will then be drawn on the screen. Drawing a circle is slower than drawing the other shapes but, even so, takes only a few seconds. You will see that the circle has its centre at the centre of the screen. Probably you will not want it in this position, but it is easy to move it elsewhere, as explained later. When the circle is finished, the ' BCDMPRTZ?' message appears again, showing that the computer is waiting for your next command. If you want to draw another circle, repeat the routine described above, choosing a different circle number.

You can choose a different colour too, if you wish, and the new circle can have any radius.

The routine for drawing rectangles is very similar to that for drawing circles. You press 'R' select the number of the rectangle, then its colour. After this you are asked to key in the width of the rectangle in screen units. You are also asked to key in the required height of the rectangle, also in screen units. If you want a square, key in the same value as you did for width. After you have entered these two values, the rectangle appears at high speed. One of its corners is at the centre of the screen. It can be moved to any position on the screen, and can be rotated to different angles, as will be explained later.

Polygons are generally used for the more complicated and unusual shapes. A polygon is a shape with any number of straight sides. The program allows the polygons to have up to twelve sides, though each can have a different number of sides. A room with a chimney-piece and a bay window has twelve sides, so a polygon is ideal for drawing the outline of such a room. At the other extreme, a three-sided polygon (or triangle) might represent a corner-cupboard. It is even possible to draw a one-sided polygon, in other words, a straight line (you could use one of these for drawing the door opening into a room, or for drawing a fence in a garden. Taking things to extremes, how about a zero-sided polygon? This appears as a single dot of colour, which might prove useful, though it is rather wasteful of the memory space provided for storing the details of polygons.

Drawing polygons is different from drawing circles and rectangles. You first press 'P' and are then asked to select the polygon number and its colour as explained previously. Then you will see a small dot of the specified colour appear near the top left corner of the screen. The message 'Use arrow/ copy keys' appears in the message area at the bottom of the screen. Drawing a polygon is done by telling the computer where its corners are to be. As you press the arrow keys (the four to the top right of the keyboard, normally used for editing) a line will be drawn starting from the initial coloured dot (Fig. 1.1.). Take care not to press the BREAK key! The lines that you draw will not necessarily be the sides of polygon in its final form. Their main purpose is to give you a rough outline. So you need not worry if you are bad at drawing and your line wanders erratically about the screen. The computer will straighten everything later. Also, as you can see, the keys can only draw horizontal or vertical lines, but you may well want some diagonal ones too.




Fig 1.1. Beginning to draw a polygon

As stated above, the essential point is to mark the positions of the corners of the polygon. When the end of the line has reached a point where you want a corner to be, press the COPY key. At once you will see a marker appear. The marker is a short diagonal line, its lower end marking the spot where the corner is to be located on the screen (Fig. 1.2). The marker flashes in contrasting colours, so that you can easily pick it out. Having placed one marker, use the arrow keys to move on to the next corner. Note that you must mark each corner in turn, as if




Figure 1.2. The placing of markers for a 5-sided polygon

you were walking completely around the room with one hand on the wall, or walking around the boundaries of the garden.

There is no need to complete the final side of the polygon by drawing a line; just mark all its corners and the computer will later join them all up.

When you have marked all the corners, press the space-bar. The screen clears completely, removing all your wandering lines and any other shapes you have already plotted. Then all the shapes reappear: first the circles, followed quickly by the rectangles, and finally the polygon you have just marked (Fig. 1.3). It now consists of straight lines joining the marked points (*without* markers).




Fig 1.3. The polygon as drawn by the computer, given the markers placed as in Fig. 1.2.

If you use all twelve markers in drawing a polygon, the clearing and redrawing operations described above take place without your having to press the space-bar first.

If you want to draw a room or garden accurately, you may find it hard to gauge the exact places where its corners should be. Figure I .4 shows how to position the corners accurately. Follow the steps in this figure from A to G. Although we have dealt with circles, rectangles and polygons in that order, you can of course, select any shape whenever 'BCDMPRSTZ' appears.

Deleting and bringing back shapes

If you make a mistake when drawing a circle, rectangle or polygon, it is easy to delete it. If you are in the middle of drawing a polygon, and make a mistake in *marking a corner* (making a mistake in your




Fig. 1.4. Technique for placing a marker exactly below or to one side of a marker which is far away. When displayed by the computer, the figure above is a perfect rectangle.

wandering line does not matter!), press the space-bar. The computer redraws your faulty polygon very neatly, but this is easy to get rid of. When the 'BCDMRSTZ?' message appears, you are able to delete anything on the screen.

To delete a shape, press 'D'. The next message to appear is 'Which shape? (CPR)' Press 'C', 'P' or 'R' depending on whether you want to delete a circle, a rectangle or a polygon. Then you are asked '*Which one? (0-9)' Key in the number of the shape you want deleted and it instantly disappears. It is very useful to keep a note of the numbers of the shapes you have defined so that you know which one you want to delete. You can then draw another shape to replace the one you have just deleted, giving it the same number as before.

Deletion can be useful in other ways. You may be planning a room and want to know what it will look like without a given item of furniture. Just delete the shape concerned.

The 'bring-back' command is the opposite of delete. It brings back to the screen any shape which has been deleted, provided that you have not drawn another shape of the *same number* in the meantime. If circle number 5 was an apple tree and you wanted to see what the garden plan would be like without it, you could delete it. Then you could put it back again by using the 'bringback' command. This command is set up in exactly the same way as the delete command, except that you press 'B' to start with.

Moving a shape

Moving a shape is initiated by pressing *M' in response to the *BCDMPRSTZ?' message. You are asked to key in the kind of shape (' C' ,P' or ' R') and its number, as with the delete and ' bringback' commands. These commands soon become easy to key in, especially as you do not have to press the RETURN key. For example, you key ' D' ,C" In rapid succession to ' DeletCircle 2' To ' MovRectangle 6' the keys are ' M' , ' R' , ' 6' .

When you have keyed ' M' ; C" ,If for example, to ' MovCircle 1' , the message ' Usarrow/copy keys' appears. Pressing the arrow keys moves the shape a short distance across the screen in the direction indicated by the arrow. Press the key very briefly when moving circles. As might be expected, moving a circle for an appreciable distance is a relatively slow operation, but moving the other shapes is very fast. You can move the shapes so that one overlaps the other. You can even move a shape entirely off the screen. This is rather like clearing the furniture out of the room. Provided that you can remember off which side of the screen you moved it, you can bring the shape back on to the screen again later.

Pressing the COPY key causes the shape to rotate. It rotates clockwise. Forty key-presses are needed to turn it completely around. The point about which it rotates is the centre of the drawing area. If you want to rotate a shape that is near the edge of the screen, you may find it easier to move it to the centre of the screen, rotate it the desired amount and then move it back to its position at the edge of the screen. The COPY key has no effect on circles.

When you have finished moving or rotating the shape, press the space-bar. This returns you to the ' BCDMPRTZ?' message.

Changing scale

The scale of the drawing can be altered at any time. When you press ' S'the message ' width?' appears. Respond to this by typing in the number of Screen units that you wish the screen to be, from 16H to right. Initially it is 640, but you can change this to any positive integer value you wish. Type in the number, and press RETURN. There is a short delay while the micro recalculates all the values it has stored. Then the screen clears and the shapes are plotted to the new scale.

If the number you select is greater than the scale previously in use, the objects are displayed smaller than before. As an experiment, having drawn some shapes with the standard 640 units as screen width, change to a screen width of, say, 3000. The shapes reappear on

a very small scale clustered in the centre of the screen. You might, for example, have planned this layout of the furniture on the patio; now you can change to another scale (key in a large number) to plan the remainder of the garden. If you change the scale back to its initial value, all the shapes return to their original sizes.

Keying in a small scale number has the effect of magnifying the shapes (i.e. larger sizes). Then you see only those shapes which were previously close to the centre of the screen. You see only parts of other shapes, or perhaps they are completely off-screen. As an example, you can enlarge the shape which represents the dining table, and 'lay the table' with plates, then shrink it again to its initial size. The plates may appear only as dots, but you can easily change the scale again if you want to see them better. If you delete a shape, change scale and then 'bring back' the shape, the shape is plotted to the new scale in its correct position.

Tidying the screen

Moving a shape around the screen, or deleting a shape which overlaps another shape is bound to destroy the continuity of some of the lines. The simple answer to this problem is to use the 'tidy' command. Just press 'T' to make the screen clear and have the shapes perfectly redrawn. Also use 'T' when you have just loaded a plan from tape. This will cause all previously drawn shapes to appear on the screen.

Saving on tape

When you key 'Z' the message 'FILENAME?' appears. Key in the name which you want to give to the plan. Then press RETURN. The next message is 'RECORD then RETURN'. Wind the tape so that a fresh piece of tape is ready at the recorder head. Then press RECORD and PLAY on the recorder. Finally press RETURN. The plan is saved when the motor of the recorder stops and the computer emits its customary beep. You are then offered the chance to make a backup copy. If you decided to do this, press 'Y'. The routine given above is then repeated. If not, press 'N' which returns you to the command message, 'BCDMRSTZ?'

Your plan will have remained on the screen during the saving routine. If you wish, you can modify it. To begin a new plan, press ESCAPE and then rerun the program. Press 'T' to recover the shapes.

A simple measuring 'tape'

When you draw circles or rectangles, you know the exact measurements of each (in screen units). This is because the computer has asked you to key in their sizes and you have, or should have, made a written note of their sizes. It is not as easy to draw polygons to exact sizes because you are never asked to key in their dimensions. You just draw them out on the screen. You can estimate their dimensions roughly by knowing that the screen normally measures 640 units across and 480 units deep.

If you want to draw polygons to an exact size, make yourself a screen scale to help you lay out the polygon accurately. Define a rectangle which is a given number of screen units long. For example, you might define one that is 100 units long. It need be only five units wide. It is better to have two or three, or even more, such screen scales. If, for example, you want one side of a polygon to be 400 units long, begin by setting one scale with one of its ends where one end of the side of the polygon is to be. Then use two other scales of the same length, placing them end to end, and 'stepping' them along a distance of 400 units. Leave one of these scales in position to mark where the other end of the side of the polygon is to be. If you want to measure other sides accurately, repeat this process, using more scales if necessary, until the positions of all essential corners have been marked. Press the space-bar when all scales are in position. Then key 'P' to draw a polygon, using these scales to show you where to place the markers. When this is finished, delete all scales and tidy the screen. If you have used several rectangles as scales and have now finished drawing polygons, you can, of course, redefine these rectangles to other sizes, and colours for use as furniture or other objects.

Keying in

The array in which the details of polygons is stored, `P%`, has three dimensions (see line 30). Whenever you come across this array in the listing, make sure that you type all three numbers or variables that occur between the brackets. The VDU commands in statements in lines 130 and 140 have an assortment of commas and semicolons separating the numbers. It is important to type these in exactly as

given in the listing. Note especially that the statement in line 130 ends with a semicolon, whereas the statement in line 140 ends with a number.

The *FX 4,1 call in line 150 disables the editing function of the ' arrowkeys and the COPY key. This is done so that they can be used in the routines for drawing polygons and for moving and rotating shapes. If you run the program and then perhaps find a typing error that has to be corrected, these keys will not work in their usual way for editing. To restore their editing function, type:

```
*FX 4,0
```

and press RETURN. It is a good idea to program one of the Electron's function keys to restore the editing function. The way to do this is to type:

```
*KEY2 "*FX 4,0|M"
```

Then press RETURN. This programs function key 2 (marked ' f2 on its front), but you can program any other function key in the same way. To use the function key, press the CAPS LK/FUNCT key and the ' f2key at the same time. To get this to work the cursor must be at the left of the screen, next to the ' f2prompt mark. If it is at any other position on the line you get a ' Syntaxerror' message. So if you have tried using the arrow or COPY keys when disabled (which merely moves the cursor along the line) press RETURN first, to get the cursor back to the beginning of the line. Then use the function key as described above.

Line 1760 uses the arc-cosine function, ACS. This is not the same as the more frequently-used ASCII code function, ASC. Line 1770 uses the constant PI (which has the value 3.14159265), not a variable called PI.

Program design

- 20-40 Initialising the screen.
- 50-120 Tape input routine.
- 130-140 Defining a graphics window and a text window.
- 150 Disabling the arrow and COPY keys.
- 160-170 Setting colours.

16 *Practical Programs for the Electron*

- 180-280 Waiting for your command.
- 290-350 Drawing a circle. 360-430 Drawing a rectangle.
- 440-480 Bringing back a shape.
- 490-530 Deleting a shape.
- 540-630 Moving a circle.
- 640-690 Moving or rotating a rectangle.
- 700-750 Moving or rotating a polygon.
- 760-860 Placing markers when drawing a polygon.
- 870-1010 Recalculating values when changing scale.
- 1020-1150 Redrawing all shapes (used by Tidy, after placing polygon markers and after changing scale).
- 1160-1250 Saving on tape routines.
- 1260-1310 PROCno: requests the number of the shape to be drawn, moved, etc.
- 1320-1360 PROCsp: requests the code letter for the kind of shape to be drawn, moved, etc.
- 1370-1410 PROCsz: requests the size of a shape (radius, etc.) or of screen width when changing scale.
- 1420-1470 PROCcol: requests the colour for a shape.
- 1480-1540 PROCc: Draws a circle.
- 1550-1600 PROCr: Draws a rectangle.
- 1610-1660 PROCp: Draws a polygon.
- 1670-1720 PROCm: Detects presses on the arrow-keys, when moving a shape and when drawing a polygon.
- 1730-1790 PROCrot: Recalculates the values of the co-ordinates when a shape is rotated.

Points of interest

The details of the circles, rectangles and polygons are stored in three arrays, C%(), R%() and P%(), respectively. The colour of each shape is stored as a number from 1 to 7. This number has 8 added to it if the shape is 'active' That is to say, the shape has been drawn and has not been deleted. When a shape is deleted, 8 is subtracted from the stored value for colour. This means the details of the shape are still held in the array allowing it to be 'brought back' when required. When this is done, the colour value is increased by 8 again.

The program

```
10 REM ** SPACE PLAN **
```

```

20 MODE 2
30 DIM C%(9,3),R%(9,10),P%(9,12,1)
40 S=2
50 REPEAT:INPUT TAB(0,5)"PLANS ON TAP
E? (Y/N) ";:KEY$=GET$:UNTIL KEY$="Y" OR
KEY$="N"
60 IF KEY$="N" THEN 130
70 INPUT'"NAME OF FILE? "N$:N$=LEFT
$(N$,7)
80 X=OPENIN N$
90 FOR J=0 TO 9:FOR K=0 TO 3:INPUT#X,
C%(J,K):NEXT:NEXT
100 FOR J=0 TO 9:FOR K=0 TO 10:INPUT#X
,R%(J,K):NEXT
110 FOR J=0 TO 9:FOR K=0 TO 12:FOR L=0
TO 1:INPUT#X,P%(J,K,L):NEXT:NEXT:NEXT
120 CLOSE#X
130 VDU 24,0;64;1279;1023;
140 VDU 28,0,31,19,29
150 *FX 4,1
160 GCOL 0,128:CLG
170 COLOUR 4:COLOUR 134
180 REPEAT:CLS
190 PRINT"BCDMPRSTZ? ";
200 KEY$=GET$
210 UNTIL INSTR("BCDMPRSTZ",KEY$)
220 IF KEY$="B" THEN 440
230 IF KEY$="D" THEN 490
240 IF KEY$="M" THEN 540
250 IF KEY$="P" THEN 760
260 IF KEY$="S" THEN 870
270 IF KEY$="T" THEN 1020
280 IF KEY$="Z" THEN 1160
290 PROCno
300 PROCcol
310 IF KEY$="R" THEN 360
320 PROCsz("radius")
330 C%(N%,0)=SZ%:C%(N%,1)=C%+8:C%(N%,2
)=639:C%(N%,3)=544
340 PROCc(N%)
350 GOTO 180
360 PROCsz("width")
370 R%(N%,0)=SZ%
380 PROCsz("height")

```

18 Practical Programs for the Electron

```
390 R%(N%,1)=SZ%:R%(N%,2)=C%+8:R%(N%,3)
)=639+S*R%(N%,0):R%(N%,4)=544
400 R%(N%,5)=R%(N%,3):R%(N%,6)=544-S*R
%(N%,1):R%(N%,7)=639:R%(N%,8)=R%(N%,6)
410 R%(N%,9)=639:R%(N%,10)=544
420 PROCr(N%)
430 GOTO 180
440 PROCsp:PROCno
450 IF KEY$="C" AND C%(N%,1)<8 THEN C%
(N%,1)=C%(N%,1)+8:PROCC(N%)
460 IF KEY$="R" AND R%(N%,2)<8 THEN R%
(N%,2)=R%(N%,2)+8:PROCr(N%)
470 IF KEY$="P" AND P%(N%,0,0)<8 THEN
P%(N%,0,0)=P%(N%,0,0)+8:PROCP(N%)
480 GOTO 180
490 PROCsp:PROCno
500 IF KEY$="C" AND C%(N%,1)>7 THEN C%
(N%,1)=C%(N%,1)-8:PROCC(N%)
510 IF KEY$="R" AND R%(N%,2)>7 THEN R%
(N%,2)=R%(N%,2)-8:PROCr(N%)
520 IF KEY$="P" AND P%(N%,0,0)>7 THEN
P%(N%,0,0)=P%(N%,0,0)-8:PROCP(N%)
530 GOTO 180
540 PROCsp:PROCno
550 CLS:PRINT"Use arrow/copy keys"
560 IF KEY$="R" THEN 640
570 IF KEY$="P" THEN 700
580 X=C%(N%,2):Y=C%(N%,3)
590 REPEAT:P$=INKEY$(0):PX=C%(N%,2):PY
=C%(N%,3)
600 PROCm
610 IF X<>PX OR Y<>PY THEN C%(N%,1)=C%
(N%,1)-8:PROCC(N%):C%(N%,2)=X:C%(N%,3)=Y
:C%(N%,1)=C%(N%,1)+8:PROCC(N%)
620 UNTIL P$=" "
630 GOTO 180
640 REPEAT:P$=INKEY$(0):X=0:Y=0
650 PROCm
660 IF X<>0 OR Y<>0 THEN R%(N%,2)=R%(N
%,2)-8:PROCr(N%):R%(N%,2)=R%(N%,2)+8:FOR
R=3 TO 9 STEP 2:R%(N%,R)=R%(N%,R)+X:R%(
N%,R+1)=R%(N%,R+1)+Y:NEXT:PROCr(N%)
670 IF P$=CHR$(135) THEN R%(N%,2)=R%(N
%,2)-8:PROCr(N%):R%(N%,2)=R%(N%,2)+8:FOR
```

```

R=3 TO 9 STEP 2:X=R%(N%,R):Y=R%(N%,R+1)
:PROCrOt:R%(N%,R)=X:R%(N%,R+1)=Y:NEXT:PR
OCr(N%)
  680 UNTIL P$=" "
  690 GOTO 180
  700 REPEAT:P$=INKEY$(0):X=0:Y=0
  710 PROCm
  720 IF X<>0 OR Y<>0 THEN P%(N%,0,0)=P%
(N%,0,0)-8:PROCP(N%):P%(N%,0,0)=P%(N%,0,
0)+8:FOR R=1 TO P%(N%,0,1):P%(N%,R,0)=P%
(N%,R,0)+X:P%(N%,R,1)=P%(N%,R,1)+Y:NEXT:
PROCP(N%)
  730 IF P$=CHR$(135) THEN P%(N%,0,0)=P%
(N%,0,0)-8:PROCP(N%):P%(N%,0,0)=P%(N%,0,
0)+8:FOR R=1 TO P%(N%,0,1):X=P%(N%,R,0):
Y=P%(N%,R,1):PROCrOt:P%(N%,R,0)=X:P%(N%,
R,1)=Y:NXT:PROCP(N%)
  740 UNTIL P$=" "
  750 GOTO 180
  760 PROCno:PROCCol
  770 GCOL 0,C%:P%=1:P%(N%,0,0)=C%+8
  780 CLS:PRINT"Use arrow/copy keys"
  790 X=10:Y=900:PLOT 69,X,Y
  800 REPEAT:P$=INKEY$(0):GCOL 0,C%
  810 IF P$=CHR$(135) THEN P%(N%,P%,0)=X
:P%(N%,P%,1)=Y:P%=P%+1:GCOL 0,C%+8:PLOT
1,32,32:PLOT 0,-32,-32:GOTO 840
  820 PROCm
  830 PLOT 5,X,Y
  840 UNTIL P%=13 OR P$=" "
  850 P%(N%,0,1)=P%-1
  860 GOTO 1020
  870 PROCsz("width"):PS=S:S=960/SZ%:F=S
/PS
  880 FOR J=0 TO 9
  890 C%(J,2)=639+(C%(J,2)-639)*F
  900 C%(J,3)=544+(C%(J,3)-544)*F
  910 NEXT
  920 FOR J=0 TO 9
  930 FOR K=3 TO 9 STEP 2
  940 R%(J,K)=639+(R%(J,K)-639)*F
  950 R%(J,K+1)=544+(R%(J,K+1)-544)*F
  960 NEXT:NEXT
  970 FOR J=0 TO 9

```

20 Practical Programs for the Electron

```
  980 FOR K=1 TO 12
  990 P%(J,K,0)=639+(P%(J,K,0)-639)*F
1000 P%(J,K,1)=544+(P%(J,K,1)-544)*F
1010 NEXT:NEXT
1020 CLG
1030 FOR J=0 TO 9
1040 IF C%(J,1)<8 THEN 1060
1050 PROCc(J)
1060 NEXT
1070 FOR J=0 TO 9
1080 IF R%(J,2)<8 THEN 1100
1090 PROCr(J)
1100 NEXT
1110 FOR J=0 TO 9
1120 IF P%(J,0,0)<8 THEN 1140
1130 PROCp(J)
1140 NEXT
1150 GOTO 180
1160 CLS
1170 INPUT"FILE NAME? "N$:N$=LEFT$(N$,7
)
 1180 X=OPENOUT N$
 1190 FOR J=0 TO 9:FOR K=0 TO 3:PRINT#X,
C%(J,K):NEXT:NEXT
 1200 FOR J=0 TO 9:FOR K=0 TO 10:PRINT#X
,R%(J,K):NEXT:NEXT
 1210 FOR J=0 TO 9:FOR K=0 TO 12:FOR L=0
TO 1:PRINT#X,P%(J,K,L):NEXT:NEXT:NEXT
 1220 CLOSE#X
 1230 REPEAT:CLS:PRINT"BACKUP? (Y/N) ";
KEY$=GET$:UNTIL KEY$="Y" OR KEY$="N"
 1240 IF KEY$="Y" THEN 1180
 1250 GOTO 180
 1260 DEF PROCno
 1270 REPEAT:CLS
 1280 PRINT"Which one? (0-9) ";
 1290 N$=GET$:N%=VAL(N$)
 1300 UNTIL N%>=0 AND N%<10
 1310 ENDPROC
 1320 DEF PROCsp
 1330 REPEAT:CLS
 1340 PRINT"Which shape (CPR) ";
 1350 KEY$=GET$:UNTIL INSTR("CPR",KEY$)
 1360 ENDPROC
```

```

1370 DEFPROCsz(M$)
1380 REPEAT:CLS
1390 PRINT"What ";M$;:INPUT"? "A$
1400 SZ%=VAL(A$):UNTIL SZ%>0
1410 ENDPROC
1420 DEF PROCcol
1430 REPEAT:CLS
1440 PRINT"Colour? (1-7) ";
1450 C$=GET$:C%=VAL(C$)
1460 UNTIL C%>0 AND C%<8
1470 ENDPROC
1480 DEF PROCc(N)
1490 R%=S*C%(N,0)
1500 FL=5:IFC%(N,1)<8 THEN FL=7:GOTO 15
20
1510 GCOL 0,C%(N,1)-8
1520 PLOT 4,C%(N,2),C%(N,3)+R%
1530 FORL=1 TO 30:PLOT FL,C%(N,2)+R%*SIN(L*PI/15),C%(N,3)+R%*COS(L*PI/15):NEXT
1540 ENDPROC
1550 DEF PROCr(N)
1560 F=5:IFR%(N,2)<8 THEN F=7:GOTO 1580
1570 GCOL 0,R%(N,2)-8
1580 PLOT 4,R%(N,9),R%(N,10)
1590 FOR L=1 TO 4:PLOT F,R%(N,1+2*L),R%(N,2+2*L):NEXT
1600 ENDPROC
1610 DEF PROCp(N)
1620 F=5:IF P%(N,0,0)<8 THEN F=7:GOTO 1
640
1630 GCOL 0,P%(N,0,0)-8
1640 PLOT 4,P%(N,P%(N,0,1),0),P%(N,P%(N,0,1),1)
1650 FOR L=1 TO P%(N,0,1):PLOT F,P%(N,L,0),P%(N,L,1):NEXT
1660 ENDPROC
1670 DEF PROCm
1680 IF P$=CHR$(136) THEN X=X-10
1690 IF P$=CHR$(137) THEN X=X+10
1700 IF P$=CHR$(138) THEN Y=Y-10
1710 IF P$=CHR$(139) THEN Y=Y+10
1720 ENDPROC
1730 DEF PROCrot
1740 IF X=639 AND Y=544 THEN ENDPROC

```

22 Practical Programs for the Electron

```
1750 X=X-639:Y=Y-544:Q=SQR(X*X+Y*Y)
1760 A=ACS(Y/Q):IF X<0 THEN A=2*PI-A
1770 A=A+PI/20:IFA>2*PI THEN A=A-2*PI
1780 X=639+Q*SIN(A):Y=544+Q*COS(A)
1790 ENDPROC
```

Variations

There is memory to spare to allow you to increase the number of shapes that can be defined. Simply increase the first dimension (the '9') of the array on line 30, to one less than the number of shapes you require. The '9' lines 90 (for circles), 100 (for rectangles), 110 (for polygons), 880 (for circles), 920 (for rectangles), 970 (for polygons), 1030 (for circles), 1070 (for rectangles), 1110 (for polygons), 1190 (for circles), 1200 (for rectangles) and 1210 (for polygons) should be increased too. You will also need to adapt PROCno to allow you to key in numbers larger than 9. This will mean altering the message and substituting an INPUT statement, and altering the upper limit in line, 1300. If you have different numbers of the three kinds of shapes, the procedure will need to take this into account. If you have a printer attached to the Electron, a routine to print out the display on paper would be a valuable extension of the program.