

7

Cashflow

This is a spreadsheet program, specially designed to help you plan your family finances. It helps you forecast your spending for a year or slightly more. You tell the computer how you want it to work out monthly income or your monthly expenditure on each item of your home budget, and it automatically does this for the whole planning period. This is helpful in itself, but the program really shows its merits when you decide to alter the way one or more of the items is calculated. You simply type in the alteration and the computer recalculates that item and any other items that are related to it. The example which follows later explains in more detail what CASHFLOW can do.

Being able to plan accurately, you are more easily able to foresee periods when you are likely to be short of cash. It is better to know about this in advance, so that you can cut down expenses in time, or can arrange for extra income or a loan. On the other hand, if there is something you want to buy now and CASHFLOW shows that you will have enough spare cash in the foreseeable future, you can confidently make your purchase immediately.

But please remember that the computer is basing its calculations on information that you feed into it. If you give it the wrong information, or do not clearly tell what to do with that information, the results it hands back to you could possibly be wrong. Treat the program as an aid to intelligent money management. Treat the program as an aid to intelligent money management.

Using the program

The first thing to decide is the period for which you wish to work out your cashflow. This can be for any number of months from 2 to 14. A

14-month period allows you to plan for a full year ahead and also to see how your finances are likely to be as the next year begins. The first question the program asks is 'Starting from which month?'. Key in a number between 1 and 12 in reply to this question. The number is the number of the month *in the calendar year* i.e. 1 = January, 2 = February, and so on. Next you are asked how many months you wish the cashflow to cover. Key in a number from 2 to 14.

Finally, you are asked 'How many items?'. This is the number of separate items of income or expenditure that you wish to include in the cashflow. If you are in doubt, it is better to ask for more items than you think you will need. It is all too easy to remember an essential item when you are halfway through the program, but at this stage it is not possible to increase the number of items. On the other hand, if you have a spare item or two that is subsequently not used, it makes no difference to the calculations.

The screen clears and then the column headings of the spreadsheet appear. These look like the top two rows of Fig. 7.1. To set up this example, the starting month selected was '4' (April), and the number of months chosen was 14. The cashflow runs from April of one year until May of the next year. Note that the columns are numbered from 1 to 14. The first April is column 1, and the second May is column 14. These are the numbers used to identify each month from now on.

At this stage, there is a letter of the alphabet at the left end of each row of the table. There is one lettered row for each item you have chosen to work with.

Further input to the program is at the bottom left of the screen. The first question you will see there is 'Item A Name:'. Type in the name of the first item, noting that only the first seven letters of the name will be used to label the spreadsheet. In the example of Fig. 7.1, the first name typed in was 'INCOME'. As soon as RETURN is pressed, the name appears on row A. The next question is 'initial value (0—9999)'. On most occasions you key in zero, but the example given later will explain in more detail what you should do at this stage.

Now the computer wants to know how it is to calculate the values in the spreadsheet for each month of the period. You first see the message 'Period 1-'. You may wish the computer to perform the same calculation for all months of the period. If so, type '14' (or whatever number of months your cashflow is covering). You may wish the calculation to cover only, say, the first four months of the period. In this case, type 4. If you want the calculation to be for the first month only, type 1.

	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec	Jan	Feb	Mar	Apr	May
	1	2	3	4	5	6	7	8	9	10	11	12	13	14
A INCOME	550	550	550	550	550	550	550	550	550	600	600	600	600	600
B HOUSE	150	100	100	100	100	100	150	100	100	100	100	100	150	100
C CAR	65	65	65	95	65	65	65	65	65	65	65	20	20	20
D FOOD	180	180	180	180	180	180	180	180	180	180	180	180	180	180
E FUEL	15	15	15	15	15	15	30	30	30	30	30	30	15	15
F HOLIDAY	108	132	156	176	0	24	40	62	84	112	146	180	209	245
G SAVINGS	367	460	534	567	620	662	639	664	684	774	831	901	927	991
H CLOTHES	73	92	107	113	124	132	128	133	137	149	166	180	185	198

Fig. 7.1. Typical household cashflow.

As soon as you have typed a number and pressed RETURN, an '=' appears after it. This is the computer's way of inviting you to tell it what calculation is to be done for the month or months you have specified. 'Doing a calculation' sounds impressive, but often only a very simple expression is to be typed in here. If this is your first run through, type in a number (e.g. 50). Otherwise see the example given later.

If you have specified a period up to the last month of the cashflow, you will then be asked to repeat the above entries for item B and subsequent items in order. If you have specified, say, the first four months as your first period, the next message to appear will be 'Period 5-' .Now you can select a second period, in which the basis for calculation can be entirely different. As before, this period can run to the end of the cashflow (type the number of the last month, e.g. 14), or for a few months, or for only one month. The computer will continue asking you to key in details until you have told it what to do for every month of the cashflow. Before proceeding to the next item it always asks 'ALLOK?' At this stage, check your entries and if there are mistakes, key 'N' The area at the bottom of the screen clears. The item name in the table clears also. You can then start again to re-enter item name, initial value, and the calculations for as many periods as you want. When all is correct, key 'Y' to go to the next item.

When all items have been keyed in, you will see columns of figures appearing on the spreadsheet at high speed. At the end, it will look something like Fig. 7. 1 . This is your cashflow chart, which you can now examine at leisure. It may have taken you some time to key in the details for each item, but most of the work is done now. From now on, making changes to one row at a time takes little effort, yet can produce the most interesting results. At the bottom of the screen you will see the message 'Amendtem? (Key letter)' If you want to alter row C, for example, press key 'C' You will then be able to alter any detail for that row, except its name. The procedure is just the same as that for entering the details initially, as already described.

As soon as the amended details have been entered, all figures in the cashflow are recalculated and displayed. You will be able to spot the cursor flashing down the columns, with new values replacing old ones where necessary. Once again you are asked 'Amendtem?' There is no end to the program; you can carry on amending items indefinitely and each time you will see the latest version of your cashflow chart.

A cashflow example

The best way to learn how to use CASHFLOW is to follow this example. To begin with, the Smith family selected April as the ' From' month, and chose a 14 month period. They asked for only eight items, to keep the example fairly simple. The screen cleared and the display showed the column headings and row letters (not the item names) as in the figure. Apart from this, the rows and columns of the table were blank.

They were then asked to key in items A to H. These are the details they keyed in:

Item A: Name: INCOME: initial value = 0 (more about initial value later).

Their income is £550 monthly, until the new year when it will rise to £600. In this example, we are assuming that this is net income after deduction of income tax, national insurance contributions and other amounts deducted at source. The details were entered as follows: in response to ' Period1-' ,they keyed ' 9'(the column number for December). After the ' equalsign, they keyed the amount ' 550'While entering this information, the line ofthe display goes through these stages:

```

Period 1-
Period 1-5      (then press RETURN)
Period 1-5 =
Period 1-5 = 550 (then press RETURN)

```

All that they had to type in were the ' 5'and the ' 550'Typing in a single numerical value after the ' equalsmakes the computer put that amount in every column of the period specified when it displays the cashflow later.

After they had keyed ' 550'and pressed RETURN, and since they had not specified a period up to the end of the cashflow, the computer asked ' Period 6-' . This shows the stages in the next response:

```

Period 6-
Period 6-14     (then press RETURN)
Period 6-14 =
Period 6-14 = 600(then press RETURN)

```

All they had to type was the ' 6'and the ' 600'This completed the entries for Item A. When asked ' ALLOK?' ,the Smiths replied by keying ' Y' . Note that no figures appear in the cashflow yet.

Item B: Name: HOUSE: Initial value = 0. The expenditure was reckoned as £100 monthly for mortgage repayments, plus £50 in April and October for rates. When they had finished entering the details, the screen showed:

```
Item B Name: HOUSE
Initial value (0—9999) 0
    Period 1-1      = 150
    Period 2-6      = 100
    Period 7-7      = 150
    Period 8-12     = 100
    Period 13-13    = 150
    Period 14-14    = 100
```

(If you have more than six periods, data collection is continued at the right side of the screen.)

Item C: Name: CAR: Initial value = 0. The car expenses are £65 monthly until next January. After this the hire purchase ends and it costs only £20 a month. In July there is an extra £30 for a servicing before going on holiday. These details were keyed in just as for Item B above, the periods being 1-3, 4-4, 5-10 and 11-14.

Item D: Name: FOOD: Initial value = 0. This is even simpler, since Mrs Smith estimated she spent £180 monthly.

Item E: Name: FUEL: Initial value = 0. They spend £15 monthly in the summer, and £30, monthly in the winter.

Item E: Name: HOLIDAYS. Having entered the basic items of their cashflow, they now come to the less essential, though very important items. Holidays are financed by setting aside a certain amount each month. They have already accumulated £90 since returning from last year's holiday, so HOLIDAY begins with an initial value of 90. This is the purpose of 'initialvalue'. It lets you begin a cashflow with its amount of money in hand. If you want to do a cashflow over a period longer than 14 months, you could start a new cashflow using all the values from the last month of the previous cashflow as 'initialvalues' of the new one.

Having keyed in the initial value, they now have to work out how the holiday fund will be calculated each month. The basis for calculation they decided on is to find out how much money they have

left from their income after deducting for HOUSE, CAR, FOOD and FUEL. One eighth of this is to be put into the HOLIDAY fund. This amount might be put into a separate account at the bank, or (less safely) kept in a piggy bank. If they represent INCOME by the letter A, HOUSE by B, CAR by C, FOOD by D and FUEL by E, the amount of money left over from income after these expenses is:

$$A - B - C - D - E$$

One eighth of this is:

$$(A - B - C - D - E) / 8$$

This is just how you would type out such a calculation as a program line in BASIC.

The amount calculated so far is the amount they put into the holiday fund each month. The amount actually in the fund is that amount plus any amount already there. Since F is the letter referring to the HOLIDAY item, they represent the amount already in the fund by the corresponding lower-case letter, letter 'f'. The complete calculation of the amount in the fund becomes:

$$f + (A - B - C - D - E) / 8$$

In words, the amount in the fund is the amount already there, plus an eighth of what is left over from INCOME after deducting essential expenses. This is their holiday saving plan and they must tell the computer about it. Telling the computer is simple - just type in the equation above, exactly as printed. They want the computer to work in this way for all months up to July.

The display will show:

$$\text{Period 1 -4} = f + (A - B - C - D - E) / 8$$

It is important not to confuse capital letters and small letters. Since the display is in 80-column mode it is easier to make mistakes with certain letters. Look carefully at what you have typed before pressing RETURN.

In August, they go on holiday and happily spend everything saved:

$$\text{Period 5-5} = 0$$

Their holiday fund is exhausted (and so are the Smith family!) but, on their return, they start saving for next year's holiday according to the same plan:

$$\text{Period 6-14} = f + (A - B - C - D - E) / 8$$

This is exactly the same as for the pre-holiday period. Note that letter 'f' stands for whatever is in the fund. In month 1, the computer takes the 'initialvalue' as f. In month 2 and subsequent months it takes the value it has calculated for F in the previous month. Thus any 'small' letter stands for 'initialvalue' in month 1, and for 'previousmonth' after that.

Item G: Name: SAVINGS. This is the amount they have in the bank. At the start of the cashflow period, the initial value is £245. The amount in the bank at the end of each month may be calculated as follows:

Last month's balance + income - all expenses.

Translated into letters, this becomes:

$$g + A - B - C - D - E - (F - f) - h$$

The first six letters are easily understood. The pair (F-f) are required because the value F always represents the *total* in the holiday fund. The amount actually *put into* the fund in any given month is the difference between the amount in the fund in that month (F) and the amount in the fund in the previous month (f), that is to say (F - f). Letter 'h' refers to expenses on clothing in the previous month, which we shall deal with below.

This expression covers all months of the cashflow except August (1-4 and 6-14). In this holiday month, the value of F suddenly becomes zero. Using (F - f) gives a wrong result. Instead they must work out directly what to put into the fund in August (and spend straight away). For August (5-5) the SAVINGS item is calculated as:

$$g + A - B - C - D - E - (A - B - C - D - E) / 8 - h$$

(F - f) has been replaced by one eighth of income minus essential expenses.

Item H: Name: CLOTHES. This might be taken to include other expenses, such as household equipment, and hobbies. The Smiths decide to look at their bank account at the end of the month and spend one fifth of it on clothes (etc.) The formula for this is:

$$G/5$$

In making this entry they keyed 0 as the initial value. Each month

they spend one fifth of the amount left in the bank at the end of the previous month (h). Now we can see why 'h' appeared in the calculation for SAVINGS.

Having keyed in these details they are rewarded by the display shown in Fig. 7.1. Looking first at that crucial item, SAVINGS, they were pleased to see that there would be a substantial increase in the months ahead. There will be a slight fall in October, but it will be nothing to worry about. The amount they can spend on clothing is increasing likewise - perhaps they would not need to spend all of this allocation. The HOLIDAY fund rises to £176 in July. This, plus part of the FOOD and FUEL allowance for August, is available for holidaying, plus an eighth of A - B - C - D - E (£24). They will have over £200 to spend. Some of this can immediately be used as a deposit and they book a holiday costing up to that amount, knowing that they can afford it when the time comes. In fact, with such a good rate of saving, perhaps they could spend more. The Smiths can also see that by next April, their holiday fund will hold £245. This is better than the £90 they have this April, indicating the possibility of affording a more expensive holiday next year.

Then a thought occurs - why not sell the old car in July, when it is due to be serviced, and buy a new one? Their estimate for having a new car would put up the CAR expenditure to £120 monthly.

In response to 'Amend entry' they key 'C' then enter the following (after keying '0' for initial value):

$$1-3 = 65$$

$$4-14 = 120$$

Immediately a new cashflow chart is generated, taking this new calculation into account. Since F affects item G directly, and item H indirectly (the amount for clothing depends on how much has been saved), figures in rows F, G and H are all changed.

Before they have time to study these, another thought occurs to them. Mrs Smith has allowed £180 monthly for food, but there is inflation to think of. If they want to eat as well in 14 months time as now, she may need to spend more on food each month. Inflation is about 6% per annum. Roughly, this is a half per cent increase monthly (actually it is less, but the difference is not important). So she begins the food calculation with an initial value of £180 and inflates this by half a per cent each month. Initial value is 180 and the calculation of D is

$$I - 14 = d * 1.005$$

Each month the amount of the previous month is multiplied by 1.005. Once again the figures in the table change. This time, rows D, F, G and H are affected. Figure 7.2 shows the results of the imendments to CAR and FOOD. FOOD increases by £1 each month, which is less worrying than she thought it might be. Happily, [t lookk as if the Smith family can afford the new car. SAVINGS are not as great as was predicted by Fig. 7.1, but there is still an overall increase.

The drop in savings in October, November and December might have been very worrying if they had not done a cashflow and were relying only on monthly bank statements received at the ends of those months. CASHFLOW shows that the situation will improve when Mr Smith gets the rise in pay next January. Rather than wait for the rise before buying the car, they bought the car in time for this year' s holiday, knowing that they can afford it in the long run. It is for this kind of long-range planning that a cashflow forecast is most helpful. But remember that the forecast is only as reliable as the information you feed into the computer. If you have forgotten to tell it that you are spending £100 a month on entertaining your friends to parties, the financial future may be less rosy than it looks on the chart. Also an unexpected expense, such as wind damage to your roof, could (unless you are insured) reduce your SAVINGS balance considerably. Keep funds in reserves for emergencies, and insure yourself against those risks you cannot afford to meet.

Good things have to be paid for, a fact that is shown on the chart. The purchase of a new car means that you must spend less on your holiday, and on clothes. If you think you might prefer to put a bigger portion of your surplus cash into the holiday fund, save less and spend less on clothing, you have only to amend item F and see what affect this will have in the next 14 months. Perhaps ' $f+(A-B-C-D-E)/6$ would be a better basis for setting aside your holiday money. Perhaps ' $G/8$ ' would provide enough for clothing. Having done this, you might be able to spend even more on the car! You can continue indefinitely in this way, trying one thing, then another, until you have planned your cashflow to your satisfaction.

Other calculations

The example has used various kinds of' calculations' These have ranged from one or more constants (items A to E), a simple multiplication or division by a given factor (item H, and the inflation-

	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec	Jan	Feb	Mar	Apr	May
	1	2	3	4	5	6	7	8	9	10	11	12	13	14
A INCOME	550	550	550	550	550	550	550	550	550	600	600	600	600	600
B HOUSE	150	100	100	100	100	100	150	100	100	100	100	100	150	100
C CAR	65	65	65	120	120	120	120	120	120	120	120	120	120	120
D FOOD	181	182	183	184	185	186	187	188	189	190	191	192	193	194
E FUEL	15	15	15	15	15	15	30	30	30	30	30	30	15	15
F HOLIDAY	107	131	154	170	0	16	24	38	52	72	92	112	127	148
G SAVINGS	367	458	530	539	545	549	494	493	491	533	565	590	579	613
H CLOTHES	73	92	106	108	109	110	99	99	98	107	113	118	116	123

Fig. 7.2. The cashflow of Fig. 7.1, amended to allow for buying a new car and for inflation of food prices.

indexed calculation for D) to a long succession of additions and subtractions (items F and G). The example covers most of the kinds of calculation that you are likely to want to use, but you can use even more complex ones. The routine which handles the calculations is capable of working out any mathematical expression as in a BASIC program line. If you wanted to, you could calculate how much to spend on clothes using:

$$G/SIN(D)+LOG(e^B)$$

What this means in terms of family budgeting is hard to say! But beware! Syntax errors or invalid statements made when typing in the calculation line could cause the program to crash. Another point worth noting is that only integers are used in the results. Negative values are allowed: this is when you start to think about asking your bank manager for an overdraft! If possible, try not to let your cashflow produce values bigger than 99999, as this will upset the neat rows and columns of the display. If your cashflow runs to six figure amounts, you can afford to buy more expensive spreadsheet software.

We have assumed that you will use only reasonably sensible calculations in this program. The range of possibilities is great. We prefer not to restrict what you can key in, even though there is the risk of your keying in an unacceptable item. In this way, this fairly short program has the greatest possible flexibility in use.

One final point to consider is that the program works column by column, passing down the columns as it goes (the rapid motion of the cursor reveals the sequence). Consequently, values in a column are updated from the top downward. It may not be sensible for a calculation of a given item to include a CAPITAL letter relating to an item further down the list. This later item could possibly be recalculated as the program proceeds, making nonsense of the result worked out earlier. If you know from what you have entered that a value further down the column will not change then including it in an item further up the list is allowable. It is always in order to use lower-case letters in calculations, no matter what their position in the list may be.

Keying in

A number of the strings have spaces in them. There are examples on lines 50, 70, 500 and elsewhere. Take care to key in the correct

number of spaces, otherwise an untidy display may result.

The variable in lines 120, 170 and 310 is ' NI(letter *en' lettei ' eye'). The last statement in line 190 is:

```
1=D%(Z-1, 12)
```

The letter on the left is ' smalel' .The number after the ' Zis figure ' one' .

Program design

20-30 Initialising mode, arrays, and a variable.

40-90 Queries relating to the scope ofthe cashHow.

100-110 Displaying table with row and column labels.

120-140 Entering information until all correct.

150 Setting format for displaying values (integers, up to five figures).

160-240 Taking values for a to 1 and A to L out ofthe array, using these to evaluate the calculation; storing the result in the array and also displaying it in the table.

250 Restoring normal figure display format.

260-330 Clear bottom of screen and input data to amend calculations.

340-390 PROCqueries to display a query message, input numbers between pre-set upper and lower limits, the message being tabbed to a given screen line.

400-430 PROChead to display column headings (month names and numbers).

440-470 PROCside to display letters on left side of screen.

480-730 PROCdata to input all information relating to a given item.

740-760 PROCclear to clear the lower area of the screen.

770-790 DATA statement containing sequence ofmonth names for column headings.

Points of interest

The program uses the EVAL function (line 220) to evaluate the ' calculationexpressions keyed in as strings and stored in D\$(). The

102 Practical Programs for the Electron

results are stored in D%(), the addition of 9.5 at line 220 serving to place a rounded-off value in D%().

The program

```
10 REM ** CASHFLOW **
20 MODE 3
30 DIMD%(14,12),D$(14,12),M$(14):FA=0
40 PRINT' TAB(35)"CASHFLOW"
50 PROCqueries("Starting from which month? (1-12) ",1,12,5)
60 FOR J=1 TO M-1:READ X$:NEXT
70 PROCqueries("For how many months? (2-14) ",2,14,7)
80 NM=M:FOR J=1 TO NM:READ X$:M$(J)=X$:NEXT
90 PROCqueries("How many items? (1-12) ",1,12,9)
100 NI=M:CLS:VDU 19,0,4,0,0,0
110 PROChead:PROCside
120 FOR J=1 TO NI
130 REPEAT:PROCdata:UNTIL KEY$="Y" OR KEY$="y"
140 NEXT
150 @%=&00005
160 FOR Z=1 TO NM
170 FOR ZZ=1 TO NI
180 a=D%(Z-1,1):b=D%(Z-1,2):c=D%(Z-1,3):d=D%(Z-1,4):e=D%(Z-1,5):f=D%(Z-1,6)
190 g=D%(Z-1,7):h=D%(Z-1,8):i=D%(Z-1,9):j=D%(Z-1,10):k=D%(Z-1,11):l=D%(Z-1,12)
200 A=D%(Z,1):B=D%(Z,2):C=D%(Z,3):D=D%(Z,4):E=D%(Z,5):F=D%(Z,6)
210 G=D%(Z,7):H=D%(Z,8):I=D%(Z,9):J=D%(Z,10):K=D%(Z,11):L=D%(Z,12)
220 D%(Z,ZZ)=EVAL (D$(Z,ZZ))+.5
230 PRINT TAB(5+5*Z,2+ZZ)D%(Z,ZZ);
240 NEXT:NEXT
250 @%=10
260 PROCclear
270 FA=1
280 REPEAT
290 INPUT TAB(0,16)"Amend item ? (Key
```

```

letter)  "L$
  300 J=ASC(L$)-64:IF J>32 THEN J=J-32
  310 UNTIL J>0 AND J<NI+1
  320 PROCdata
  330 GOTO 150
  340 DEF PROCqueries(M$,LL,UL,TB)
  350 REPEAT
  360 PRINT TAB(0,TB)M$;:INPUT"  "A$
  370 M=VAL(A$)
  380 UNTIL M>LL-1 AND M<UL+1
  390 ENDPROC
  400 DEF PROChead
  410 FOR J=1 TO NM:PRINT TAB(6+5*J,1)M$
(J);:NEXT
  420 FOR J=1 TO NM:PRINT TAB(6+5*J);J;:
NEXT
  430 ENDPROC
  440 DEF PROCside
  450 VDU 31,0,3
  460 FOR J=1 TO NI:PRINT CHR$(J+64):NEX
T
  470 ENDPROC
  480 DEF PROCdata
  490 REPEAT
  500 IF FA=0 THEN PRINT TAB(2,J+2)"
  "
  510 PROCclear
  520 PRINT TAB(4,16)"Item ";CHR$(J+64);
  530 IF FA=1 THEN PRINT TAB(15,16)D$(0,
J):GOTO 570
  540 INPUT TAB(15,16)"Name:  "D$
  550 D$(0,J)=LEFT$(D$,7)
  560 PRINT TAB(2,J+2)D$(0,J)
  570 PROCqueries("Initial value (0-9999
)  ",0,9999,17)
  580 D%(0,J)=M
  590 X=2:Y=17:FM=1
  600 REPEAT
  610 Y=Y+1:IF Y=24 THEN X=41:Y=16
  620 REPEAT
  630 PRINT TAB(X,Y)"Period ";FM;:INPUT"
-"LM$
  640 LM=VAL(LM$)
  650 UNTIL LM>FM-1 AND LM<NM+1

```

104 Practical Programs for the Electron

```
660 INPUT TAB(X+12,Y)" = "E$
670 FORK=FM TO LM:D$(K,J)=E$:NEXT
680 FM=LM+1
690 UNTIL LM=NM
700 PRINT TAB(66,Y)"ALL OK? (Y/N)";
710 KEY$=GET$
720 UNTIL KEY$="Y" OR KEY$="y"
730 ENDPROC
740 DEF PROCclear
750 VDU 31,0,16:FOR K=1 TO 8:PRINT STR
ING$(80," ");:NEXT
760 ENDPROC
770 DATA Jan, Feb, Mar, Apr, May, Jun, Jul,
Aug, Sep, Oct, Nov, Dec
780 DATA Jan, Feb, Mar, Apr, May, Jun, Jul,
Aug, Sep, Oct, Nov, Dec, Jan
790 DATA Jan, Feb
```

Variations

The program could easily be adapted to handle data of other kinds, including non-integer values. You would need to change the column headings, several of the messages and probably the number of rows and columns.